

ACTIVIDAD PRÁCTICA: TELEMETRÍA Y TELECONTROL

Estudiante

Guadalupe Sosa Acosta | guadalupe.sosa@estudiantes.utec.edu.uy | 5.660.478-1

UTEC: Instituto Regional Sur Oeste

Fray Bentos – Río Negro - Uruguay

Fecha de realización

25/04/2026

Desarrollo

Opción 2: Control de Invernadero Inteligente

Para la resolución de esta práctica, se seleccionó la Opción 2: Control de Invernadero Inteligente. Se utilizó un potenciómetro como sensor analógico para simular la humedad del suelo y un LED conectado a un pin digital para actuar como electroválvula. El microcontrolador utilizado fue un ESP8266 y se utilizó un Broker MQTT local.

El código fuente completo, detallando la lógica de conexión Wi-Fi, gestión de MQTT y el desafío extra, se encuentra disponible en el Apéndice A.

Tabla 1. Pinout.

Fuente: Elaboración Propia.

Potenciómetro	
Vcc	3.3 V
GND	GND
Output	A0
LED	
Ánodo	D1
Cátodo	GND



Fig.1 Diagrama de conexiones del sistema. Fuente: Elaboración Propia.




Fig.2 Conexión del sistema. Fuente: Elaboración Propia.

Tabla 2. Tópicos.

Fuente: Elaboración Propia.

Tópicos	Tipo	Descripción
invernadero/monitoreo	Publicación (Telemetría)	Pública el % de humedad y el estado actual cada 12s.
invernadero/comando	Suscripción (Telecontrol)	Recibe las órdenes: AUTO, MANUAL_ON o MANUAL_OFF.

Se optó por una arquitectura de tópico unificado para la telemetría. Al enviar el porcentaje de humedad junto con el estado del sistema en un solo mensaje, se reduce el tráfico de red y se garantiza que el usuario reciba la información contextualizada; saber si la válvula está abierta porque la humedad es baja (<30%) o porque está en modo manual.

```
PS C:\Users\sosa2> mosquitto_sub -h 172.22.192.203 -t invernadero/monitoreo
78.8% [Modo: AUTO | Valvula: CERRADA]
78.8% [Modo: AUTO | Valvula: CERRADA]
78.8% [Modo: MANUAL | Valvula: CERRADA]
78.8% [Modo: MANUAL | Valvula: CERRADA]
78.8% [Modo: MANUAL | Valvula: CERRADA]
78.8% [Modo: MANUAL | Valvula: CERRADA]
100.1% [Modo: MANUAL | Valvula: CERRADA]
0.5% [Modo: MANUAL | Valvula: CERRADA]
100.1% [Modo: MANUAL | Valvula: CERRADA]
51.4% [Modo: MANUAL | Valvula: CERRADA]
51.4% [Modo: AUTO | Valvula: CERRADA]
14.6% [Modo: AUTO | Valvula: CERRADA]
0.5% [Modo: AUTO | Valvula: ABIERTA]
42.9% [Modo: AUTO | Valvula: ABIERTA]
```

Fig.3 Consola de Monitoreo. Fuente: Elaboración Propia.

La figura 3 corresponde a una captura de la PowerShell (suscriptor), donde se observa la recepción de datos. Se valida entonces que el sistema informa correctamente los cambios de estado cuando la humedad varía.

```
PS C:\Users\sosa2> mosquitto_pub -h 172.22.192.203 -t "invernadero/comando" -m "AUTO"
PS C:\Users\sosa2> mosquitto_pub -h 172.22.192.203 -t "invernadero/comando" -m "MANUAL_ON"
PS C:\Users\sosa2> mosquitto_pub -h 172.22.192.203 -t "invernadero/comando" -m "MANUAL_OFF"
PS C:\Users\sosa2> mosquitto_pub -h 172.22.192.203 -t "invernadero/comando" -m "AUTO"
```

Fig.4 Consola de Comandos. Fuente: Elaboración Propia.

La figura 4 muestra el envío de mensajes desde la terminal hacia el broker. Se evidencia cómo el sistema procesa los comandos AUTO, MANUAL_ON y MANUAL_OFF para tomar o ceder el control del hardware.

```
78.9% [Modo: AUTO | Valvula: ABIERTA]
78.8% [Modo: AUTO | Valvula: CERRADA]
78.8% [Modo: AUTO | Valvula: CERRADA]
78.8% [Modo: AUTO | Valvula: CERRADA]
Mensaje recibido: AUTO
78.8% [Modo: AUTO | Valvula: CERRADA]
Mensaje recibido: MANUAL_ON
Mensaje recibido: MANUAL_OFF
78.8% [Modo: MANUAL | Valvula: CERRADA]
78.8% [Modo: MANUAL | Valvula: CERRADA]
78.8% [Modo: MANUAL | Valvula: CERRADA]
```

Fig.5 Monitor Serial. Fuente: Elaboración Propia.

En el monitor serial se puede visualizar el registro de eventos internos: la conexión exitosa al SSID, la dirección IP asignada y el feedback de los comandos recibidos en la función callback, parte de esto se observa en la figura 5.

El video demostrativo del funcionamiento, donde se comprueba el cumplimiento de la telemetría periódica, el telecontrol y la lógica híbrida del desafío extra, se encuentra referenciado en el Apéndice B.

Apéndices

Apéndice A. Código

```
#include <ESP8266WiFi.h>
#include <PubSubClient.h>

const char* ssid = "Sosa";
const char* password = "12345678";
const char* mqtt_server = "172.22.192.203";

WiFiClient espClient;
PubSubClient client(espClient);

#define LED_PIN D1
#define POT_PIN A0

unsigned long lastMsg = 0;
const long intervalo = 12000; // 12 segundos según consigna
float humedad = 0;
bool modoAuto = true; // Variable para el desafío extra

void setup_wifi() {
  Serial.begin(115200);
  WiFi.mode(WIFI_STA);
  WiFi.begin(ssid, password);
  while (WiFi.status() != WL_CONNECTED) {
    delay(500);
    Serial.print(".");
  }
  Serial.println("\nWiFi conectado");
}

void callback(char* topic, byte* payload, unsigned int length) {
  String mensaje = "";
  for (int i = 0; i < length; i++) { mensaje += (char)payload[i]; }

  Serial.print("Mensaje recibido: ");
  Serial.println(mensaje);

  if (mensaje == "AUTO") {
    modoAuto = true;
  }
  else if (mensaje == "MANUAL_ON") {
    modoAuto = false;
    digitalWrite(LED_PIN, HIGH);
  }
  else if (mensaje == "MANUAL_OFF") {
    modoAuto = false;
    digitalWrite(LED_PIN, LOW);
  }
}

void reconnect() {
  while (!client.connected()) {
    if (client.connect("ESP8266Client")) {
      client.subscribe("invernadero/comando"); // Tópico de control
    } else {
      delay(5000);
    }
  }
}
```

```
void setup() {
  pinMode(LED_PIN, OUTPUT);
  digitalWrite(LED_PIN, LOW);
  setup_wifi();
  client.setServer(mqtt_server, 1883);
  client.setCallback(callback);
}

void loop() {
  if (!client.connected()) { reconnect(); }
  client.loop();

  unsigned long now = millis();

  if (now - lastMsg > intervalo) {
    lastMsg = now;
    int valorADC = analogRead(POT_PIN);
    humedad = (valorADC / 1023.0) * 100.0;

    // Lógica de texto para el estado
    String modoTxt = (modoAuto) ? "AUTO" : "MANUAL";
    String valvulaTxt = (digitalRead(LED_PIN) == HIGH) ? "ABIERTA" : "CERRADA";

    String reporte = String(humedad, 1) + "% [Modo: " + modoTxt + " | Valvula: " + valvulaTxt + "]";

    // Publica todo en un solo tópico
    client.publish("invernadero/monitoreo", reporte.c_str());
    Serial.println(reporte);
  }

  // Desafío Extra
  if (modoAuto) {
    if (humedad < 30.0) {
      digitalWrite(LED_PIN, HIGH);
    } else {
      digitalWrite(LED_PIN, LOW);
    }
  }
}
```

Apéndice B. Video

Link: https://drive.google.com/file/d/100YiXyucvZpJSS5yZ-22y5jg4-NpVG_T/view?usp=sharing
